Ch.5 Slender plane beam element for large displacements and rotations

Consider an arbitrary element of a planar beam structure, which, in the unloaded state is in the reference position C_0 (Fig. 5.1) defined by the coordinates of the nodal points *i* and *j*.



Fig. 5.1

After loading, the beam was moved to position C_1, and the displacement components of the nodal points that arose during this movement were arranged into a vector

$$\mathbf{u}_e = \begin{bmatrix} u_i & v_i & \varphi_i & u_j & v_j & \varphi_j \end{bmatrix}^T \tag{5.1}$$

The removed parts of the structure act with force resultants at the end points of the beam, organized into a vector of the element's internal nodal forces

$$\mathbf{q}_e = \begin{bmatrix} Q_{ix} & Q_{iy} & M_i & Q_{jx} & Q_{jy} & M_j \end{bmatrix}^T \tag{5.2}$$

The procedure for determining the basic matrices of the element is analogous to that used when solving the same problem for a plane truss element (Ch.4). We need to express the terms $\mathbf{q}_e$ as functions of the components $\mathbf{u}_e$ from relation (4.6)

$$\mathbf{q}_e = \frac{\partial U_e}{\partial \mathbf{u}_e} \tag{5.3}$$

This allows the assembly of the equilibrium equations of the nodes and the tangential element matrix from the relation

$$\mathbf{K}_T^e = \frac{\partial \mathbf{q}_e}{\partial \mathbf{u}_e} \tag{5.4}$$

The tangential element matrices can then be combined into the resulting tangential matrix of the structure and used in an iterative process to solve the entire structural deformation.

When determining the strain energy of the beam, it is advantageous to use the local coordinate system of the element $\overline{x}$, $\overline{y}$, which is tied to the movement of the element, in which the nodal force components represent the axial force $N = N_j = -N_i$ and transverse forces T_i *and,* T_j . The nodal bending moments M_i , M_j are identical in both coordinate systems.

We proceed with the assumption that we are dealing with a slender beam in which the energy of the shear stresses can be neglected. Thus, the transverse forces do not contribute to the deformation of the beam, and the local vector of the element deformation forces has only three components:

$$\overline{\mathbf{q}}_e = \begin{bmatrix} N & M_i & M_j \end{bmatrix}^T \tag{5.5}$$

Because we do not consider external loads on the element (external loads will be applied only to the nodes, which we have virtually separated from the element), the transverse forces are equal in magnitude $T = T_i = -T_j$; of course, the nodal moments of the element are linked by the moment equilibrium condition with the transverse force

$$T_i = \left(M_i + M_j\right)/L \tag{5.6}$$

In the local coordinate system, the relationships between the nodal forces and displacements (Fig. 5.1), which contribute to the deformation of the beam, follow the equations known from the linear stiffness matrix of the element

$$\overline{\mathbf{q}}_e = \begin{bmatrix} N \\ M_i \\ M_j \end{bmatrix} = \begin{bmatrix} \dfrac{E_0 S_0}{L_0} & 0 & 0 \\ 0 & \dfrac{4E_0 I_0}{L_0} & \dfrac{2E_0 I_0}{L_0} \\ 0 & \dfrac{2E_0 I_0}{L_0} & \dfrac{4E_0 I_0}{L_0} \end{bmatrix} \begin{bmatrix} L - L_0 \\ \overline{\varphi}_i \\ \overline{\varphi}_j \end{bmatrix} = \mathbf{D}\overline{\mathbf{u}}_e \tag{5.7}$$

where E_0 is the modulus of elasticity of the material, S_0 is the cross-sectional area, and I_0 is the quadratic moment of the beam cross-section.

The internal nodal forces perform work during beam deformation, and the magnitude of this work is identical to the strain energy accumulated in the deformed element

$$U_e = U_N + U_M = \frac{1}{2} N\left(L - L_0\right) + \frac{1}{2}\left(M_i \overline{\varphi}_i + M_j \overline{\varphi}_j\right) \tag{5.8}$$

Using (5.7), the nodal forces of the element in (5.8) can be eliminated, thus allowing the use of the deformation formulation of the element via the following relationships:

$$U_N = \frac{1}{2} N\left(L - L_0\right) = \frac{1}{2}\sigma S_0\left(L - L_0\right) = \frac{1}{2} E_0 \varepsilon_{ing} S_0\left(L - L_0\right) = \frac{1}{2} E_0 S_0 L_0 \varepsilon_{ing}^2 \tag{5.9}$$

$$U_M = U_{M_i} + U_{M_j} = \frac{1}{2} M_i \overline{\varphi}_i + \frac{1}{2} M_j \overline{\varphi}_j = \frac{1}{2}\left(\frac{4E_0 I_0}{L_0}\overline{\varphi}_i + \frac{2E_0 I_0}{L_0}\overline{\varphi}_j\right)\overline{\varphi}_i + \frac{1}{2}\left(\frac{2E_0 I_0}{L_0}\overline{\varphi}_i + \frac{4E_0 I_0}{L_0}\overline{\varphi}_j\right)\overline{\varphi}_j \tag{5.10}$$

Quantities ε_{ing} , $\overline{\varphi}_i$ and $\overline{\varphi}_j$ include the global nodal displacements of the element, and if we want to apply (5.3) and (5.4) it is necessary to express these relationships. For simplicity, it is advantageous to search for the vector of nodal forces in the form

$$\mathbf{q}_e = \mathbf{q}_N + \mathbf{q}_{M_i} + \mathbf{q}_{M_j} = \frac{\partial U_N}{\partial \mathbf{u}_e} + \frac{\partial U_{M_i}}{\partial \mathbf{u}_e} + \frac{\partial U_{M_j}}{\partial \mathbf{u}_e} \tag{5.11}$$

Therefore, we can derive the element energies separately. The calculation $\mathbf{q}_N$ is analogous to the calculation of this vector for the truss element (Ch.4) differences arise only from the fact that here, we use the engineering strain of the beam instead of Green's strain

$$\mathbf{q}_N = \frac{\partial U_N}{\partial \mathbf{u}_e} = E_0 S_0 L_0 \varepsilon_{ing} \frac{\partial \varepsilon_{ing}}{\partial \mathbf{u}_e} = \sigma S_0 L_0 \frac{\partial \varepsilon_{ing}}{\partial \mathbf{u}_e} = N L_0 \frac{\partial \left(\frac{L - L_0}{L_0} \right)}{\partial \mathbf{u}_e} = N \frac{\partial L}{\partial \mathbf{u}_e} \tag{5.12}$$

The length of the element in the current configuration is

$$L = \sqrt{\left(X_{ji} + u_{ji} \right)^2 + \left(Y_{ji} + v_{ji} \right)^2}$$

and from (5.12) we obtain

$$\mathbf{q}_N = N \left[-\cos\beta \quad -\sin\beta \quad 0 \quad \cos\beta \quad \sin\beta \quad 0 \right]^T \tag{5.13}$$

where

$$\cos\beta = \frac{X_{ji} + u_{ji}}{L}, \; \sin\beta = \frac{Y_{ji} + v_{ji}}{L}$$

The variables with double indices express the differences in the respective quantities at nodes *j* and *i*.

The derivation of the remaining two columns of the nodal force vector (5.11) is not presented here because of its length (see e.g. [1]); its final form is

$$\mathbf{q}_e = \mathbf{B}^T \overline{\mathbf{q}}_e = \begin{bmatrix} -\cos\beta & -\frac{\sin\beta}{L} & -\frac{\sin\beta}{L} \\ -\sin\beta & \frac{\cos\beta}{L} & \frac{\cos\beta}{L} \\ 0 & 1 & 0 \\ \cos\beta & \frac{\sin\beta}{L} & \frac{\sin\beta}{L} \\ \sin\beta & -\frac{\cos\beta}{L} & -\frac{\cos\beta}{L} \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} N \\ M_i \\ M_j \end{bmatrix} \tag{5.14}$$

In this relation, it is still necessary to determine the dependence of nodal forces N, M_i and M_j on the global displacements of the element, that is, to express $L - L_0$, $\overline{\varphi}_i$ and $\overline{\varphi}_j$ in equations (5.7). Expressing the axial change in the length of the beam is straightforward because

$$L_0 = \sqrt{X_{ji}^2 + Y_{ji}^2}; \qquad L = \sqrt{\left(X_{ji} + u_{ji} \right)^2 + \left(Y_{ji} + v_{ji} \right)^2}$$

For the end rotations according to Fig. 5.1, it holds that

$$\overline{\varphi}_i = \varphi_i - \alpha; \qquad \overline{\varphi}_j = \varphi_j - \alpha \tag{5.15}$$

Further from the figure we get

$$\sin(\alpha + \delta) = \sin\beta; \qquad \cos(\alpha + \delta) = \cos\beta \tag{5.16}$$

where $sin\,\delta = Y_{ji} / L_0$, $cos\,\delta = X_{ji} / L_0$. By expanding the known trigonometric relationships for the sum of two angles, we obtain two equations, from which we obtain:

$$sin\,\alpha = \frac{1}{LL_0}\left(X_{ji}v_{ji} - Y_{ji}u_{ji}\right); \qquad cos\,\alpha = \frac{1}{LL_0}\left[X_{ji}\left(X_{ji}+u_{ji}\right)+Y_{ji}\left(Y_{ji}+v_{ji}\right)\right] \tag{5.17}$$

and for the desired angle it holds that

$$\alpha = arctan\left(\frac{X_{ji}v_{ji} - Y_{ji}u_{ji}}{X_{ji}\left(X_{ji}+u_{ji}\right)+Y_{ji}\left(Y_{ji}+v_{ji}\right)}\right) \tag{5.18}$$

From the known vector of the element's nodal forces, it is now possible to determine the element's tangential matrix according to Eq. (9.4). We present the results in the simplest form

$$\mathbf{K}_T^e = \mathbf{K}_M^e + \mathbf{K}_G^e = \mathbf{B}^T\mathbf{D}\mathbf{B} + \mathbf{K}_{GN}^e + \mathbf{K}_{GM}^e \tag{5.19}$$

where

$$\mathbf{K}_{GN}^e = \frac{N}{30L}\begin{bmatrix} 36\,sin^2\beta & -18\,sin\,2\beta & -3L\,sin\,\beta & -36\,sin^2\beta & 18\,sin\,2\beta & -3L\,sin\,\beta \\ & 36\,cos^2\beta & 3L\,cos\,\beta & 18\,sin\,2\beta & -36\,cos^2\beta & 3L\,cos\,\beta \\ & & 4L^2 & 3L\,sin\,\beta & -3L\,cos\,\beta & -L^2 \\ & & & 36\,sin^2\beta & -18\,sin\,2\beta & 3L\,sin\,\beta \\ & SYM & & & 36\,cos^2\beta & -3L\,cos\,\beta \\ & & & & & 4L^2 \end{bmatrix},$$

$$\mathbf{K}_{GM}^e = \frac{T}{L}\begin{bmatrix} sin2\beta & -cos2\beta & 0 & -sin2\beta & cos2\beta & 0 \\ & sin2\beta & 0 & cos2\beta & sin2\beta & 0 \\ & & 0 & 0 & 0 & 0 \\ & & & sin2\beta & -cos2\beta & 0 \\ & SYM & & & -sin2\beta & 0 \\ & & & & & 0 \end{bmatrix}$$

The material stiffness matrix in (5.19) $\mathbf{K}_M^e = \mathbf{B}^T\mathbf{D}\mathbf{B}$ expresses the material and dimensional stiffness of the element; it is essentially the stiffness matrix of the element in the local coordinate system transformed into the global system, and it represents the standard stiffness matrix for the equilibrium position of the element. The geometric stiffness matrix of the element is formed by the sum of the matrix $\mathbf{K}_{GN}^e$ (the influence of the axial force on the stiffness of the beam) and $\mathbf{K}_{GM}^e$ (the influence of the nodal moments).

Example 5.1

For the planar beam structure in the figure, the horizontal and vertical components of the displacement of node 3 were calculated when given: a = 1000 mm, square cross-sectional area A_0 = 100 mm^2, E_0 = 200 000 MPa, I_0 = $10^4/12=833.33$ mm^4 , and F = 50 N.



Program *Mathematica 7* is constructed for single-purpose use, only for the needs of this example; its modification to compute multiple-element examples is straightforward, but this is not its primary intention. Above all, it serves as an illustrative demonstration of the method for using matrices of the nonlinear beam element.

- Vector *Vectorq* with given problem parameters (a E_0 S_0 I_0) and variables $u2, v2, t2, u3, v3, t3$

```
q = Vectorq[1000, 200000, 100, 833.33, u2, v2, t2, u3, v3, t3];
```

- Calculation of the internal nodal force vector using the *Vektorqe* and *Adícia VnútUzlSíl*

```
Vectorq[a_, E0_, S0_, I0_, u2_, v2_, t2_, u3_, v3_, t3_] :=
  Module[{},
  q = Table[0, {6}, {1}];
  qe1 = Vectorqe[0, 0, 0, a, 0, 0, 0, u2, v2, t2, E0, S0, I0];
  q = NodalForcAddit[qe1, {0, 0, 0, 1, 2, 3}, q];
  qe2 = Vectorqe[0, a, a, a, u2, v2, t2, u3, v3, t3, E0, S0, I0];
  q = NodalForcAddit[qe2, {1, 2, 3, 4, 5, 6}, q];
  Return[q]];

Vectorqe[xi_, yi_, xj_, yj_, ui_, vi_, tetai_, uj_, vj_, tetaj_,
  E0_, S0_, I0_] := Module[{},
   xji = xj - xi; yji = yj - yi;
   uji = uj - ui; vji = vj - vi;
   xu = xji + uji; yv = yji + vji;
   L0 = Sqrt[xji^2 + yji^2];
   L2 = xu^2 + yv^2; L = Sqrt[L2];
   cbeta = xu / L; sbeta = yv / L; e = (L - L0) / L0;
   calfa = xji * xu + yji * yv; salfa = xji * vji - yji * uji;
   alfa = N[Chop[ArcTan[calfa, salfa]]];
   filok1 = Chop[tetai - alfa]; filok2 = Chop[tetaj - alfa];
   NN = E0 * S0 * e;
   Mi = (2 * E0 * I0 / L0) * (2 * filok1 + filok2);
   Mj = (2 * E0 * I0 / L0) * (filok1 + 2 * filok2);
   T = (Mi + Mj) / L;
   qN = NN * {-cbeta, -sbeta, 0, cbeta, sbeta, 0};
   qM = {-T * sbeta, T * cbeta, Mi, T * sbeta, -T * cbeta, Mj};
   qe = qN + qM;
   qe = Transpose[{qe}];
  Return [qe]];
```

- Addition of element internal force contributions using the element code number (kod)

```
NodalForcAddit[qe_, kod_, p_] :=
    Module[{}, q = p;
     neldof = 6;
     For[i = 1, i ≤ neldof, i++, ii = kod[[i]];
      If[ii > 0, q[[ii, 1]] += qe[[i, 1]]]]; Return[q]];
```

- Subroutine for calculating the tangential stiffness matrix of a body with given problem parameters (a, E_0, A_0, I_0) and variables $u2, v2, t2, u3, v3, t3$

```
K = TangMatr[1000, 200 000, 100, 1000, u2, v2, t2, u3, v3, t3];

 TangMatr[a_, E0_, S0_, I0_, u2_, v2_, t2_, u3_, v3_, t3_] :=
   Module[{}, K = Table[0, {6}, {6}];
    Ke1 = TangMatrElem[0, 0, 0, a, 0, 0, 0, u2, v2, t2, E0, S0, I0];
    K = ElemMatrAddit[Ke1, {0, 0, 0, 1, 2, 3}, K];
    Ke2 = TangMatrElem[0, a, a, a, u2, v2, t2, u3, v3, t3, E0, S0, I0];
    K = ElemMatrAddit[Ke2, {1, 2, 3, 4, 5, 6}, K];
    Return[K]];

 TangMatrElem[xi_, yi_, xj_, yj_, ui_, vi_, tetai_, uj_, vj_,
    tetaj_, E0_, S0_, I0_] := Module[{},
    KM = MaterMatr[xi, yi, xj, yj, ui, vi, tetai, uj, vj, tetaj,
      E0, S0, I0];
    KG = GeoMatr[xi, yi, xj, yj, ui, vi, tetai, uj, vj, tetaj, E0, S0, I0];
    Ke = KM + KG;
    Return[Ke]];

 MaterMatr[xi_, yi_, xj_, yj_, ui_, vi_, tetai_, uj_, vj_, tetaj_,
    E0_, S0_, I0_] := Module[{},
    xji = xj - xi; yji = yj - yi;
    uji = uj - ui; vji = vj - vi;
    xu = xji + uji; yv = yji + vji;
    L0 = Sqrt[xji^2 + yji^2];
    LL = xu^2 + yv^2; L = Sqrt[LL];
    k1 = E0 * S0 / L0;  k2 = E0 * I0 / L0;
    cosb = xu / L; sinb = yv / L;
    B = {{-cosb, -sinb, 0, cosb, sinb, 0},
      {-sinb / L, cosb / L, 1, sinb / L, -cosb / L, 0},
      {-sinb / L, cosb / L, 0, sinb / L, -cosb / L, 1}};
    M = {{k1, 0, 0}, {0, 4 k2, 2 k2}, {0, 2 k2, 4 k2}};
    BT = Transpose[B];
    BTM = Inner[Times, BT, M];
    KM = Inner[Times, BTM, B];
    Return[KM]];
```

```
GeoMatr[xi_, yi_, xj_, yj_, ui_, vi_, tetai_, uj_, vj_, tetaj_,
   E0_, S0_, I0_] :=
  Module[{},
   xji = xj - xi; yji = yj - yi;
   uji = uj - ui; vji = vj - vi;
   xu = xji + uji; yv = yji + vji;
   L0 = Sqrt[xji^2 + yji^2];
   L2 = xu^2 + yv^2; L = Sqrt[L2]; LL = L * L;
   cb = xu / L; sb = yv / L; e = (L - L0) / L0;
   calfa = xji * xu + yji * yv; salfa = xji * vji - yji * uji;
   alfa = Chop[ArcTan[calfa, salfa]];
   filok1 = Chop[tetai - alfa]; filok2 = Chop[tetaj - alfa];
   NN = E0 * S0 * e;
   Mi = (2 * E0 * I0 / L0) * (2 * filok1 + filok2);
   Mj = (2 * E0 * I0 / L0) * (filok1 + 2 * filok2);
   T = (Mi + Mj) / L;
   ssb = sb * sb; ccb = cb * cb;
   s2b = 2 * sb * cb; c2b = ccb - ssb;
   KN =  {{36 ssb, -18 s2b, -3 L * sb, -36 ssb, 18 s2b, -3 L * sb},
          {-18 s2b, 36 ccb, 3 L * cb, 18 s2b, -36 ccb, 3 L * cb},
          {-3 L * sb, 3 L * cb, 4 LL, 3 L * sb, -3 L * cb, -LL},
          {-36 ssb, 18 s2b, 3 L * sb, 36 ssb, -18 s2b, 3 L * sb},
          {18 s2b, -36 ccb, -3 L * cb, -18 s2b, 36 ccb, -3 L * cb},
          {-3 L * sb, 3 L * cb, -LL, 3 L * sb, -3 L * cb, 4 LL}} * NN / (30 L);
   KMo =  {{s2b, -c2b, 0, -s2b, c2b, 0},
          {-c2b, -s2b, 0, c2b, s2b, 0},
          {0, 0, 0, 0, 0, 0},
          {-s2b, c2b, 0, s2b, -c2b, 0},
          {c2b, s2b, 0, -c2b, -s2b, 0},
          {0, 0, 0, 0, 0, 0}} * T / L;
   KG = KN + KMo;

ElemMatrAddit[Ke_, kod_, Km_] :=
  Module[{}, K = Km;
   NVE = 6;
   For[i = 1, i ≤ NVE, i ++, ii = kod[[i]];
    For[j = i, j ≤ NVE, j ++, jj = kod[[j]];
     If[ii > 0 && jj > 0, K[[jj, ii]] = K[[ii, jj]] += Ke[[i, j]]]];];
   Return[K]];
```

- Formulation of the system of equilibrium equations with the loading force 50 N

```
R2x[{u2_, v2_, t2_, u3_, v3_, t3_}] = q[[1]][[1]];
R2y[{u2_, v2_, t2_, u3_, v3_, t3_}] = q[[2]][[1]];
M2[{u2_, v2_, t2_, u3_, v3_, t3_}]  = q[[3]][[1]];
R3x[{u2_, v2_, t2_, u3_, v3_, t3_}] = q[[4]][[1]];
R3y[{u2_, v2_, t2_, u3_, v3_, t3_}] = q[[5]][[1]] + 50;
M3[{u2_, v2_, t2_, u3_, v3_, t3_}]  = q[[6]][[1]];
R⃗[{u2_, v2_, t2_, u3_, v3_, t3_}] = {R2x[{u2, v2, t2, u3, v3, t3}],
   R2y[{u2, v2, t2, u3, v3, t3}], M2[{u2, v2, t2, u3, v3, t3}],
   R3x[{u2, v2, t2, u3, v3, t3}], R3y[{u2, v2, t2, u3, v3, t3}],
   M3[{u2, v2, t2, u3, v3, t3}]};
```

- Program for solving a system of nonlinear equations

```
NewtonRaphson[tol_, max_] :=
  Module[{norma = 1, i = 0},
   u0⃗ =  {0., 0., 0., 0., 0., 0.}; Print["u⃗0 = ", u0⃗];
   Ktan[{u2_, v2_, t2_, u3_, v3_, t3_}] = K;
   While[And[i < max, norma > tol],
    u1⃗ = u0⃗ - (Inverse[Ktan[u0⃗]]).R⃗[u0⃗]; Print["u⃗"i+1, " = ", u1⃗];
    norma = Sqrt[(u1⃗ - u0⃗).(u1⃗ - u0⃗)];
    If[norma < tol, Print["r⃗"i+1, " = ", MatrixForm[R⃗[u1⃗]]]];
    u0⃗ = u1⃗;
    i = i + 1;]];
```

- Launching the iterative solution of the problem.
  Results of the iterative process ($u_2, v_2, \varphi_2, u_3, v_3, \varphi_3$) and the vector of unbalanced forces after meeting the convergence criterion are as follows:

```
Iteracia = NewtonRaphson[0.001, 20];
```

$\vec{u}_0 = \{0., 0., 0., 0., 0., 0.\}$

$\vec{u}_1 = \{150.001, -0.0025, -0.300001, 150.001, -400.004, -0.450002\}$

$\vec{u}_2 = \{148.368, -11.0702, -0.299899, 76.8439, -382.465, -0.44989\}$

$\vec{u}_3 = \{153.355, -11.8184, -0.300205, 77.332, -394.46, -0.438907\}$

$\vec{u}_4 = \{153.333, -11.828, -0.300209, 77.242, -394.443, -0.438851\}$

$\vec{u}_5 = \{153.314, -11.8249, -0.300176, 77.2434, -394.389, -0.438767\}$

$\vec{u}_6 = \{153.314, -11.825, -0.300177, 77.2436, -394.39, -0.438767\}$

$$\vec{r}_6 = \begin{pmatrix} 0.0000450263 \\ -0.000023483 \\ -0.00243064 \\ -0.0000454847 \\ 0.000018681 \\ -0.000825871 \end{pmatrix}$$

The displacement components of node No. 3 are: $u_3 =$ 77.24 mm, and $v_3 =$ - 394.39 mm. The second line (1st iteration) presents the results of the linear solution to the problem.

We also entered the example into the ANSYS Mechanical APDL program, and the results are presented in the attached video.

Using the ANSYS program, we obtained the following displacement components for node No. 3: $u_3 = 81.37$ mm; $v_3 = -377.73$ mm. Minor deviations from the previous calculation were caused by the fact that the element 188 is formulated differently in ANSYS. (The BEAM188 is suitable for analyzing slender to moderately stubby/thick beam structures. The element is based on Timoshenko beam theory, which includes shear-deformation effects.)

When displaying the solution with only two elements, we obtained only a straight-line connection of the three displaced nodes. In the real numerical analysis of beam (frame) structures using FEM, it is necessary to divide them into more than one element. After such a calculation, we obtained realistic results for vertical node displacements in the example:



[1] Felippa, C.A.: Nonlinear Finite Element Methods, University of Colorado, Boulder, 2001